// File: LearnETH.sol



// SPDX-License-Identifier: MIT

//   /$$                                               /$$$$$$$$ /$$$$$$$$ /$$   /$$
//  | $$                                              | $$_____/|__  $$__/| $$  | $$
//  | $$        /$$$$$$   /$$$$$$   /$$$$$$  /$$$$$$$ | $$         | $$   | $$  | $$
//  | $$       /$$__  $$ |____  $$ /$$__  $$| $$__  $$| $$$$$      | $$   | $$$$$$$$
//  | $$      | $$$$$$$$  /$$$$$$$| $$  \__/| $$  \ $$| $$__/      | $$   | $$__  $$
//  | $$      | $$_____/ /$$__  $$| $$      | $$  | $$| $$         | $$   | $$  | $$
//  | $$$$$$$$|  $$$$$$$|  $$$$$$$| $$      | $$  | $$| $$$$$$$$   | $$   | $$  | $$
//  |________/ \_______/ \_______/|__/      |__/  |__/|________/   |__/   |__/  |__/

// @DEV: QAZIPOLO.ETH // ------------------------------------------------------------- //                                                                              

pragma solidity ^0.8.17;

import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Strings.sol";
import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "@openzeppelin/contracts/utils/Base64.sol";

contract LearnETH is ERC721, Ownable {
    using SafeMath for uint256;
    using Strings for uint256;
    uint256 private _tokenIdCounter;
    uint256 public partnerFee;

    bool private initiated;
    address private topNodeWallet;
    uint256 private monthDuration = 2592000;
	
    uint256 public currentPartnerCourseId;
    
    // nodeType 0 = Student
    // nodeType 1 = Instructor
    // nodeType 2 = Partner
    // nodeType 3 = Special A Partner
    // nodeType 4 = Special B Partner
    // nodeType 5 = Special A (Top node)
    
    struct node {
        uint8 nodeType;
        uint256 level; // starts with 1
        uint256 position; //starts with 1
        uint256 father1;
        uint256 personalReferrer;
        uint256 balance;
        uint256 subscribedTill;
        uint256 totalRecruits;
        uint256 lastSaleAt;
    }

    uint256 private courseIndex;

    struct course {
        string courseName;
        uint256 courseFee;
        uint256 instructorTokenId;
    }

    uint256 private partnerCourseIndex;

    struct partnerCourse {
        string courseName;
        uint256 courseFee;
        uint256 instructorTokenId;
        uint256 balance;
        bool claimable;
    }

    uint256 private paymentIndex;

    struct fee {
        uint256 paidBy;
        uint256 paidTo;
        uint256 paidFor;
        uint256 paidAmount;
        uint256 paidAt;
        uint status; // 0 - locked; 1 - distributed; 2 - refunded
    }
	
	struct earnings {
        uint256 tokenId;
        uint256 amount;
    }

    // [tokenId] => struct node
    mapping(uint256 => node) public nodeDetails;
    // [level][position] => tokenId
    mapping(uint256 => mapping(uint256 => uint256)) public partnerNodeTokenMapping;
    // [courseId] => struct course
    mapping(uint256 => course) public courseDetails;
    // [courseId] => struct course
    mapping(uint256 => partnerCourse) public partnerCourseDetails;
    // [tokenId] => courseId(s)
    mapping(uint256 => uint256[]) public enrolledCourses;
    // [tokenId (Instructor)] => courseId(s)
    mapping(uint256 => uint256[]) public instructorCourses;
    // [paymentId => struct Fee
    mapping(uint256 => fee) public paymentDetails;

    constructor() ERC721("LearnETH", "LETH") {}

    event partnerAdded(uint256 partnerTokenId, uint256 referrerTokenId, uint256 father1, uint256 level, uint256 position, uint256 subscribedTill);
    event studentAdded(uint256 studentTokenId, uint256 referrerTokenId, uint256 father);
    event instructorAdded(uint256 instructorTokenId, uint256 referrerTokenId, uint256 father);
    event courseAdded(uint256 courseId, string courseName, uint256 instructorTokenId);
    event paidForCourse(uint256 paymentId, string orderId, uint256 paidBy, uint256 paidTo, uint256 paidFor, uint256 paidAmount, uint256 paidAt);
    event partnerRenewal(uint256 partnerTokenId, uint256 subscribedTill);
    event refundIssued(uint256 paymentId);
    event balanceWithdraw(uint256 withdrawnBy,uint256 withdrawnAmount);

    // feeType: 0 - studentUplineFee, 1 - studentPersonalReferrerFee, 2 - instructorFee, 3 - instructorUplineFee, 4 - instructorPersonalReferrerFee
    // feeType: 5 - instructorFeePC, 6 - instructorUplineFeePC, 7 - instructorPersonalReferrerFeePC
    // id: paymentId for type 0-4, partnerCourseId for type 5-7
    event feeDistributed(uint256 id, uint256 beneficiaryTokenId, uint256 amount, uint feeType);

    // feeType: 0 - partnerUplineFee, 1 - partnerPersonalReferrerFee
    event partnerFeeDistributed(uint256 partnerTokenId, uint256 beneficiaryTokenId, uint256 months, uint256 amount, uint feeType);

    // READ FUNCTIONS // ------------------------------------------------------------- //

    function totalSupply() public view returns(uint256) {
        uint256 supply = _tokenIdCounter;
        return supply;
    }

    // ADMIN FUNCTIONS // ------------------------------------------------------------- //

    // assigns tokenId "1" and nodeType "5" to contract Owner
    function initiatePartnerNetwork() external onlyOwner {
        require(!initiated,"Already Initited");
        require(msg.sender == owner(), "Only owner can initiate the network");
        _tokenIdCounter+=1;
        uint256 tokenId = _tokenIdCounter;
        _safeMint(msg.sender, tokenId);
        nodeDetails[tokenId].nodeType = 5; //Special Type for parent node
        nodeDetails[tokenId].level = 1;
        nodeDetails[tokenId].position = 1;
        partnerNodeTokenMapping[1][1] = tokenId;
        initiated = true;
        
        emit partnerAdded(tokenId, 0, 0, 1, 1, 0);
        _tokenIdCounter+=5781;
    }

    //assign special status to a partner
    function assignPriviledge(uint256 _tokenId, uint8 _nodeType) external onlyOwner {
        require(msg.sender == owner(), "Only owner can assign special priviledge");
        nodeDetails[_tokenId].nodeType = _nodeType;
    }

    // Contract owner sets TopWallet
    function setTopNodeWallet(address _walletAddress) external onlyOwner {
        topNodeWallet = _walletAddress;
    }

    function updatePartnerCourse(uint256 _partnerCourseId, string memory _courseName, uint256 _instructorTokenId) external onlyOwner {
        partnerCourseDetails[_partnerCourseId].courseName = _courseName;
        partnerCourseDetails[_partnerCourseId].instructorTokenId = _instructorTokenId;
    }

    function setCurrentPartnerCourseId(uint256 _partnerCourseId) external onlyOwner {
        currentPartnerCourseId = _partnerCourseId;
    }

    // This method allows Contract Owner to withdraw balance to topWallet
    function withdrawTopWallet() external onlyOwner{
        uint256 bal = nodeDetails[1].balance;
        require(bal > 0,"No balance to withdraw");
        nodeDetails[1].balance = 0;
        (bool sent, ) = topNodeWallet.call{value: bal}("");
        require(sent, "Failed to send Ether");
    }

    // To be scheduled for 1st of each month
    function updateCurrentPartnerCourseId() external onlyOwner {
        partnerCourseDetails[currentPartnerCourseId].claimable = true; 
        currentPartnerCourseId += 1;
    }

    // PUBLIC FUNCTIONS // ------------------------------------------------------------- //

    // REGISTER PARTNER // ------------------------------------------------------------- //

    function registerAsPartner(uint256 _referrerTokenId, uint256 _months) public payable {
        require(_exists(_referrerTokenId),"Referrer token Id is invalid");
        //require(msg.value == partnerFee,"Incorrect value");

        uint256 currentFather;
        uint256 _personalReferrer;

        if(nodeDetails[_referrerTokenId].nodeType >= 2) {
            currentFather = _referrerTokenId;
            _personalReferrer = _referrerTokenId;
        }
        else {
            currentFather = nodeDetails[_referrerTokenId].father1;
            _personalReferrer = nodeDetails[_referrerTokenId].father1;
        }

        _tokenIdCounter+=1;
        uint256 tokenId = _tokenIdCounter;
        _safeMint(msg.sender, tokenId);
        (uint256 _level, uint256 _position) = getNextNode(currentFather);
        uint256 f_position;

        if(_position.mod(3) == 0) {
            f_position = _position.div(3);
        }
        else {
            f_position = _position.div(3) + 1;
        }
        uint256 _father1 = partnerNodeTokenMapping[_level - 1][f_position];   
        nodeDetails[tokenId].nodeType = 2;
        nodeDetails[tokenId].level = _level;
        nodeDetails[tokenId].position = _position;
        nodeDetails[tokenId].personalReferrer = _personalReferrer;
        nodeDetails[tokenId].father1 = _father1;
        uint256 _duration = _months.mul(monthDuration);
        nodeDetails[tokenId].subscribedTill = block.timestamp.add(_duration);
        partnerNodeTokenMapping[_level][_position] = tokenId;
        nodeDetails[_personalReferrer].totalRecruits+=1;

        emit partnerAdded(tokenId, _personalReferrer, _father1, _level, _position, block.timestamp.add(_duration));
        distributePartnerCourseFee(tokenId, _months, msg.value);           
           
    }

    function upgradeInstructorAsPartner(uint256 _instructorTokenId, uint256 _months) public payable {
        require(msg.sender == ownerOf(_instructorTokenId), "Not the owner of tokenId");
        require(nodeDetails[_instructorTokenId].nodeType == 1, "Not an instructor");

        uint256 currentFather = nodeDetails[_instructorTokenId].father1;

        _tokenIdCounter+=1;
        uint256 tokenId = _tokenIdCounter;
        _safeMint(msg.sender, tokenId);
        (uint256 _level, uint256 _position) = getNextNode(currentFather);
        uint256 f_position;

        if(_position.mod(3) == 0) {
            f_position = _position.div(3);
        }
        else {
            f_position = _position.div(3) + 1;
        }
        uint256 _father1 = partnerNodeTokenMapping[_level - 1][f_position];   
        nodeDetails[tokenId].nodeType = 2;
        nodeDetails[tokenId].level = _level;
        nodeDetails[tokenId].position = _position;
        nodeDetails[tokenId].personalReferrer = currentFather;
        nodeDetails[tokenId].father1 = _father1;
        uint256 _duration = _months.mul(monthDuration);
        nodeDetails[tokenId].subscribedTill = block.timestamp.add(_duration);
        partnerNodeTokenMapping[_level][_position] = tokenId;
        nodeDetails[currentFather].totalRecruits+=1;

        emit partnerAdded(tokenId, currentFather, _father1, _level, _position, block.timestamp.add(_duration));   
        distributePartnerCourseFee(tokenId, _months, msg.value);  

    }

    function upgradeStudentAsPartner(uint256 _tokenId, uint256 _months) public payable {
        require(_exists(_tokenId) && nodeDetails[_tokenId].nodeType == 0, "Not a valid student token Id");
        //require(msg.value == partnerFee,"Incorrect value");
       
        uint256 currentReferrerTokenId = nodeDetails[_tokenId].personalReferrer;
        uint256 currentFather = nodeDetails[_tokenId].father1;

        (uint256 _level, uint256 _position) = getNextNode(currentFather);
        nodeDetails[_tokenId].nodeType = 2;
        nodeDetails[_tokenId].level = _level;
        nodeDetails[_tokenId].position = _position;
        
        uint256 f_position;
        if(_position.mod(3) == 0) {
            f_position = _position.div(3);
        }
        else {
            f_position = _position.div(3) + 1;
        }
        uint256 _father1 = partnerNodeTokenMapping[_level - 1][f_position];
        nodeDetails[_tokenId].father1 = _father1;
        nodeDetails[_tokenId].personalReferrer = currentFather;
        uint256 _duration = _months.mul(monthDuration);
        nodeDetails[_tokenId].subscribedTill = block.timestamp.add(_duration);
        partnerNodeTokenMapping[_level][_position] = _tokenId;
        nodeDetails[currentFather].totalRecruits+=1;
        
        emit partnerAdded(_tokenId, currentReferrerTokenId, _father1, _level, _position, block.timestamp.add(_duration));    
        distributePartnerCourseFee(_tokenId, _months, msg.value);    
    }

    // PARTNER FUNCTIONS // ------------------------------------------------------------- //

    function enrolForPartnerCourse(uint256 _tokenId, uint256 _months) public payable {
        require(msg.sender == ownerOf(_tokenId), "Not the owner of tokenId");
        require(nodeDetails[_tokenId].nodeType == 2 || nodeDetails[_tokenId].nodeType == 4,"Not a partner");
        uint256 _amount = msg.value;
        uint256 _duration = _months.mul(monthDuration);
        nodeDetails[_tokenId].subscribedTill = nodeDetails[_tokenId].subscribedTill.add(_duration);
        emit partnerRenewal(_tokenId, nodeDetails[_tokenId].subscribedTill.add(_duration));
        distributePartnerCourseFee(_tokenId, _months, _amount);
    }

    // REGISTER STUDENT // ------------------------------------------------------------- //

    function registerStudent(uint256 _referrerTokenId) public {
        require(_exists(_referrerTokenId), "Not a valid referrer Id");
        _tokenIdCounter+=1;
        uint256 tokenId = _tokenIdCounter;
        _safeMint(msg.sender, tokenId);
        uint256 f_tokenId  = _referrerTokenId;
        nodeDetails[tokenId].nodeType = 0;
        nodeDetails[tokenId].personalReferrer = _referrerTokenId;
        if(nodeDetails[_referrerTokenId].nodeType >= 2) {
            f_tokenId = _referrerTokenId;
        }
        else {
            while(nodeDetails[f_tokenId].nodeType < 2) {
                f_tokenId = nodeDetails[f_tokenId].father1;
            }
        }
        nodeDetails[tokenId].father1 = f_tokenId;
        
        emit studentAdded(tokenId, _referrerTokenId, f_tokenId);
    }

    function upgradeInstructorAsStudent(uint256 _instructorTokenId) public {
        require(msg.sender == ownerOf(_instructorTokenId), "Not the owner of tokenId");
        require(nodeDetails[_instructorTokenId].nodeType == 1, "Not an instructor");

        uint256 _personalReferrer = nodeDetails[_instructorTokenId].personalReferrer;
        uint256 _father1 = nodeDetails[_instructorTokenId].father1;

        _tokenIdCounter+=1;
        uint256 _studentTokenId = _tokenIdCounter;
        _safeMint(msg.sender, _studentTokenId);

        nodeDetails[_studentTokenId].nodeType = 0;
        nodeDetails[_studentTokenId].personalReferrer = _personalReferrer;
        nodeDetails[_studentTokenId].father1 = _father1;

        emit studentAdded(_studentTokenId, _personalReferrer, _father1);
    }

    // STUDENT FUNCTIONS // ------------------------------------------------------------- //

    function enrolForCourse(uint256 _tokenId, uint256[] memory _courseId, string memory _orderId) public payable {
        require(msg.sender == ownerOf(_tokenId), "Not the owner of tokenId");
        //require(msg.value == courseDetails[_courseId].courseFee, "Incorrect fee");
        for(uint i=0; i<_courseId.length;i++) {
            require(_courseId[i] < courseIndex, "Incorrect Course Id");
        
            enrolledCourses[_tokenId].push() = _courseId[i];
            nodeDetails[nodeDetails[_tokenId].personalReferrer].lastSaleAt = block.timestamp;
            paymentDetails[paymentIndex].paidBy = _tokenId;
            paymentDetails[paymentIndex].paidTo = courseDetails[_courseId[i]].instructorTokenId;
            paymentDetails[paymentIndex].paidFor = _courseId[i];
            paymentDetails[paymentIndex].paidAmount = msg.value;
            paymentDetails[paymentIndex].paidAt = block.timestamp;
            paymentDetails[paymentIndex].status = 0;

            emit paidForCourse(paymentIndex, _orderId, _tokenId, courseDetails[_courseId[i]].instructorTokenId, _courseId[i], msg.value, block.timestamp);
            paymentIndex+=1;
        }
    }

    function getRefund(uint256 _tokenId, uint256 _paymentId) public {
        require(msg.sender == ownerOf(_tokenId), "Invalid call");
        require(_tokenId == paymentDetails[_paymentId].paidBy, "Course not paid by the caller");
        require(paymentDetails[_paymentId].status == 0,"Already distributed or refunded");
        require(block.timestamp.sub(monthDuration) < paymentDetails[_paymentId].paidAt, "Refund eligibility duration lapsed");

        paymentDetails[_paymentId].status = 2;
        nodeDetails[_tokenId].balance = nodeDetails[_tokenId].balance.add(paymentDetails[_paymentId].paidAmount);
        emit refundIssued(_paymentId);

    } 

    // REGISTER INSTRUCTOR // ------------------------------------------------------------- //
   
    // An 'instructor' can be referred by an existing 'partner' or 'student' or 'instructor'
    // HOWEVER a 'student'/'instructor' referring an 'instructor' would not be entitled to any benefits
    // Both 'personalReferrer' and 'father1' of an instructor would be a 'partner'.

    function registerInstructor(uint256 _referrerTokenId) public {
        require(_exists(_referrerTokenId), "Not a valid referrer Id");
        _tokenIdCounter+=1;
        uint256 tokenId = _tokenIdCounter;
        _safeMint(msg.sender, tokenId);
        uint256 _node = _referrerTokenId;
        if(nodeDetails[_node].nodeType >= 2) {

            nodeDetails[tokenId].nodeType = 1;
            nodeDetails[tokenId].personalReferrer = _node;
            nodeDetails[tokenId].father1 = _node;
        }
        else {
            while(nodeDetails[_node].father1 < 2) {
                _node = nodeDetails[_node].father1;
            }

            nodeDetails[tokenId].nodeType = 1;
            nodeDetails[tokenId].personalReferrer = nodeDetails[_node].father1;
            nodeDetails[tokenId].father1 = nodeDetails[_node].father1;
        }
        
        emit instructorAdded(tokenId, _referrerTokenId, nodeDetails[tokenId].father1); 
    }

    function upgradeStudentAsInstructor(uint256 _studentTokenId) public {
        require(msg.sender == ownerOf(_studentTokenId), "Not the owner of tokenId");
        require(nodeDetails[_studentTokenId].nodeType == 0, "Not a student");

        uint256 _father1 = nodeDetails[_studentTokenId].father1;

        _tokenIdCounter+=1;
        uint256 _instructorTokenId = _tokenIdCounter;
        _safeMint(msg.sender, _instructorTokenId);

        nodeDetails[_instructorTokenId].nodeType = 1;
        nodeDetails[_instructorTokenId].personalReferrer = _father1;
        nodeDetails[_instructorTokenId].father1 = _father1;

        emit instructorAdded(_instructorTokenId, _father1, _father1);

    }

    function upgradePartnerAsInstructor(uint256 _partnerTokenId) public {
        require(msg.sender == ownerOf(_partnerTokenId), "Not the owner of tokenId");
        require(nodeDetails[_partnerTokenId].nodeType >= 2, "Not a partner");

        uint256 _personalReferrer = nodeDetails[_partnerTokenId].personalReferrer;
        uint256 _father1 = nodeDetails[_partnerTokenId].father1;

        _tokenIdCounter+=1;
        uint256 _instructorTokenId = _tokenIdCounter;
        _safeMint(msg.sender, _instructorTokenId);

        nodeDetails[_instructorTokenId].nodeType = 1;
        nodeDetails[_instructorTokenId].personalReferrer = _personalReferrer;
        nodeDetails[_instructorTokenId].father1 = _father1;

        emit instructorAdded(_instructorTokenId, _personalReferrer, _father1);
    }

    // INSTRUCTOR FUNCTIONS // ------------------------------------------------------------- //

    function addCourse(uint256 _tokenId, string memory _courseName, uint256 _courseFee) public {
        require(msg.sender == ownerOf(_tokenId), "Not the owner of tokenId");
        require(nodeDetails[_tokenId].nodeType == 1, "Only an instructor can add a course");
        courseDetails[courseIndex].courseName = _courseName;
        courseDetails[courseIndex].courseFee = _courseFee;
        courseDetails[courseIndex].instructorTokenId = _tokenId;
        instructorCourses[_tokenId].push() = courseIndex;
		emit courseAdded(courseIndex, _courseName, _tokenId);
        courseIndex+=1;
       
    }

    function unlockPayment(uint256 _paymentId) external {
        require(msg.sender == ownerOf(1) || msg.sender == ownerOf(paymentDetails[_paymentId].paidTo), "Not authorized");
        require(paymentDetails[_paymentId].status == 0,"Already distributed or refunded");
        require(paymentDetails[_paymentId].paidAmount > 0, "Free course not eligible for distribution");
        require(paymentDetails[_paymentId].paidAt.add(monthDuration) < block.timestamp,"Still in refund eligibility window");

        uint256 studentTokenId = paymentDetails[_paymentId].paidBy;
        uint256 instructorTokenId = paymentDetails[_paymentId].paidTo;
        uint256 totalFee = paymentDetails[_paymentId].paidAmount;
        uint256 studentReferrerFee = totalFee.mul(105).div(1200);
        uint256 instructorReferrerFee = totalFee.mul(45).div(1200);
        uint256 studentUplineFee = totalFee.mul(315).div(1200);
        uint256 instructorUplineFee = totalFee.mul(135).div(1200);

        nodeDetails[instructorTokenId].balance+=totalFee.div(2);
        emit feeDistributed(_paymentId, instructorTokenId, totalFee.div(2), 2);
        // Does this require eligibility check?
        nodeDetails[nodeDetails[studentTokenId].personalReferrer].balance+=studentReferrerFee;
        emit feeDistributed(_paymentId, nodeDetails[studentTokenId].personalReferrer, studentReferrerFee, 1);
        // Does this require eligibility check?
        nodeDetails[nodeDetails[instructorTokenId].personalReferrer].balance+=instructorReferrerFee;
        emit feeDistributed(_paymentId, nodeDetails[instructorTokenId].personalReferrer, instructorReferrerFee, 4);
        distributeUplineFee(_paymentId, studentTokenId, studentUplineFee, 0);
        distributeUplineFee(_paymentId, instructorTokenId, instructorUplineFee, 3);

        paymentDetails[_paymentId].status = 1;
    }

    function distributePartnerCourseBalance(uint256 _tokenId, uint256 _partnerCourseId) external {
        require(msg.sender == ownerOf(_tokenId), "Not authorized");
        require(_tokenId == partnerCourseDetails[_partnerCourseId].instructorTokenId,"Not the instructor of this course");
        require(partnerCourseDetails[_partnerCourseId].claimable, "Not claimable");

        nodeDetails[_tokenId].balance += partnerCourseDetails[_partnerCourseId].balance.mul(75).div(100);
        emit feeDistributed(_partnerCourseId, _tokenId, partnerCourseDetails[_partnerCourseId].balance.mul(75).div(100), 5);
        nodeDetails[nodeDetails[_tokenId].personalReferrer].balance += partnerCourseDetails[_partnerCourseId].balance.mul(75).div(1200);
        emit feeDistributed(_partnerCourseId, nodeDetails[_tokenId].personalReferrer, partnerCourseDetails[_partnerCourseId].balance.mul(75).div(1200), 7);
        distributeUplineFee(_partnerCourseId, _tokenId, partnerCourseDetails[_partnerCourseId].balance.mul(225).div(1200), 6);
        partnerCourseDetails[_partnerCourseId].balance = 0;
    }

    // GENERIC FUNCTIONS // ------------------------------------------------------------- //

    // This method allows users Partners/Students/Instructors (except TokenId 1) to withdraw their balance
    function withdrawBalance(uint256 _tokenId) public {
        require(msg.sender == ownerOf(_tokenId) && _tokenId != 1, "Not Authorized");
        uint256 bal = nodeDetails[_tokenId].balance;
        require(bal > 0,"No balance to withdraw");
        nodeDetails[_tokenId].balance = 0;

        (bool success, ) = msg.sender.call{value: bal}("");
        require(success, "Withdrawal failed.");
		emit balanceWithdraw(_tokenId, bal);
    }

    // HELPER FUNCTIONS // ------------------------------------------------------------- //

    function getNextNode(uint256 _tokenId) public view returns(uint256 _level, uint256 _position) {
        require(_exists(_tokenId) && nodeDetails[_tokenId].nodeType >=2, "Not a partner");
        uint256 level = nodeDetails[_tokenId].level;
        uint256 currentPosition = nodeDetails[_tokenId].position;
        uint256 positionLeft = currentPosition;
        uint256 positionRight = currentPosition;
        uint256 position;
        uint256 id = 1;

        while(id != 0) {  

            level = level + 1;
            positionLeft = positionLeft.mul(3).sub(2);
            positionRight = positionRight.mul(3);

            for(position = positionLeft; position<= positionRight; position++) {
                id = partnerNodeTokenMapping[level][position];
                if (id == 0) {
                    break;
                }                
            }            
        }

        return(level,position);
    }

    function estimateFeeDistribution(uint256 _paymentId) public view returns(earnings[] memory studentUpline, earnings[] memory instructorUpline) {
        uint256 studentTokenId = paymentDetails[_paymentId].paidBy;
        uint256 instructorTokenId = paymentDetails[_paymentId].paidTo;
        uint256 totalFee = paymentDetails[_paymentId].paidAmount;
        uint256 studentReferrerFee = totalFee.mul(105).div(1200);
        uint256 instructorReferrerFee = totalFee.mul(45).div(1200);
        uint256 studentUplineFee = totalFee.mul(315).div(1200);
        uint256 instructorUplineFee = totalFee.mul(135).div(1200);

        earnings[] memory _studentUpline = new earnings[](10);
        earnings[] memory _instructorUpline = new earnings[](11);

        //earnings[9] memory _studentUpline;
        // assign 0th index to student's personal Referrer followed by student upline
        _studentUpline[0] = earnings(nodeDetails[studentTokenId].personalReferrer, studentReferrerFee);

        uint256 _father1 = nodeDetails[studentTokenId].father1;
        for(uint i=1; i<=9; i++) {
            if(_father1 == 1) {break;}
            else {
                _studentUpline[i] = earnings(_father1, studentUplineFee.div(9));
            }  
            _father1 = nodeDetails[_father1].father1;      
        }

        // assign 0th index to Instructor
        // assign 1st index to instructor's personal Referrer followed by Instructor upline
        _instructorUpline[0] = earnings(instructorTokenId,totalFee.div(2));
        _instructorUpline[1] = earnings(nodeDetails[instructorTokenId].personalReferrer, instructorReferrerFee);

        _father1 = nodeDetails[instructorTokenId].father1;
        for(uint i=2; i<=10; i++) {
            if(_father1 == 1) {break;}
            else {
                _instructorUpline[i] = earnings(_father1, instructorUplineFee.div(9));
            }  
            _father1 = nodeDetails[_father1].father1;      
        } 

        return(_studentUpline, _instructorUpline);
    } 

    function isEligibleForDistribution(uint256 _tokenId, uint _depth) internal view returns(bool) {
        uint256 nodeType = nodeDetails[_tokenId].nodeType;
        if(nodeType == 0 || nodeType == 1 ) { return false;} //unlikely scenario
        else if(nodeType == 3) {return true;} //exempted from partner requirements
        else if(nodeType == 4) {
            uint256 subscribedTill = nodeDetails[_tokenId].subscribedTill;
            if(subscribedTill < block.timestamp) {return false;}
            else return true;
        }
        else {

            uint256 subscribedTill = nodeDetails[_tokenId].subscribedTill;
            uint256 totalRecruits = nodeDetails[_tokenId].totalRecruits;
            uint256 lastSaleAt = nodeDetails[_tokenId].lastSaleAt;

            if(subscribedTill < block.timestamp || lastSaleAt < block.timestamp.sub(monthDuration)) {return false;}
            else if(totalRecruits == 0 || totalRecruits <= 2 && _depth > 3 || totalRecruits <=4 && _depth > 6) {return false;}
            else return true;

        }
    }  

    function distributePartnerCourseFee(uint256 _tokenId, uint256 _months, uint256 _amount) internal {
        require(_months > 0 && _months <=12, "Invalid months count");
        //require(_amount == _months.mul(partnerCourseFee), "Incorrect fee");
        uint256 instructorSharePerMonth = (_amount).mul(30).div(100).div(_months);
        uint256 partnerShare = (_amount).mul(70).div(100);

        for(uint i=0; i< _months; i++) {
            partnerCourseDetails[currentPartnerCourseId.add(i)].balance+=instructorSharePerMonth;
        }
        
        uint256 twelth = partnerShare.div(12);
        nodeDetails[nodeDetails[_tokenId].personalReferrer].balance+=twelth.mul(3);
        emit partnerFeeDistributed(_tokenId, nodeDetails[_tokenId].personalReferrer, _months, twelth.mul(3), 1);

        distributePCUplineFee(_tokenId, twelth.mul(9), _months);
    }

    function distributeUplineFee(uint256 _paymentId, uint256 _tokenId, uint256 _amount, uint _type) internal {
        uint256 _father1 = nodeDetails[_tokenId].father1;
        uint256 _share = _amount.div(9);
        for(uint i=1; i<=9; i++) {
            if(_father1 == 1) {break;}
            else {
                if(isEligibleForDistribution(_father1,i)) {
                    nodeDetails[_father1].balance+=_share;
                    emit feeDistributed(_paymentId, _father1, _share, _type);
                    _amount-=_share;
                }
                else {
                    emit feeDistributed(_paymentId, _father1, 0, _type);  
                }
            }  
            _father1 = nodeDetails[_father1].father1;      
        }        
        nodeDetails[1].balance += _amount;
    }

    function distributePCUplineFee(uint256 _tokenId, uint256 _amount, uint _months) internal {
        uint256 _father1 = nodeDetails[_tokenId].father1;
        uint256 _share = _amount.div(9);
        for(uint i=1; i<=9; i++) {
            if(_father1 == 1) {break;}
            else {
                if(isEligibleForDistribution(_father1,i)) {
                    nodeDetails[_father1].balance+=_share;
                    emit partnerFeeDistributed(_tokenId, _father1, _months, _share, 0);
                    _amount-=_share;
                }
                else {
                    emit partnerFeeDistributed(_tokenId, _father1, _months, 0, 0);  
                }
            }  
            _father1 = nodeDetails[_father1].father1;      
        }        
        nodeDetails[1].balance += _amount;
    }

    // FUNCTIONS OVERRIDES // ------------------------------------------------------------- //

    function tokenURI(uint256 tokenId) public view override returns (string memory) {
        string memory memberType; 
        
        if(nodeDetails[tokenId].nodeType == 0) {
            memberType = "Student";
        }
        else if(nodeDetails[tokenId].nodeType == 1) {
            memberType = "Instructor";
        }
        else memberType = "Partner";

        string memory json = Base64.encode(
            bytes(
                string(
                    abi.encodePacked('{"name": "LearnETH #', Strings.toString(tokenId), '", "description": "LearnETH is a Self-governed global eLearning marketplace, where people can teach or gain skills and knowledge for a fee or for free.", "image": "ipfs://bafybeig7mbjsf2yd6q44xigohksdmokmu7jqf3c2m5d6i4ajtfluqeevee/learnETH.png" , "attributes": [{"trait_type": "Member Type", "value": "',memberType,'"}]}'
                    )
                )
            )
        );

        string memory output = string(
            abi.encodePacked("data:application/json;base64,", json)
        );
        return output;
    }
    
}
// File: @openzeppelin/contracts/utils/Base64.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        /// @solidity memory-safe-assembly
        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner nor approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}